Mechanics - Free Symmetric Top short ver 11-10-18

N. T. Gladd

Initialization: Be sure the file *NTGUtilityFunctions.m* is in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[79]:=
```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Background

This is the seventh in a series of Mathematica notebooks on classical mechanics. This series was motivated by a close reading and problem solving project I undertook in 2014. The focus of my attention was the text ***Introduction to Classical Mechanics with Problems and Solutions***, by David Morin. This is a good book from which to learn and has a great collection of problems. I purchased it and recommend that those with interests in this topic acquire it for their library. I do note that an earlier version can be found on the web. This year, when I returned to this project, I decided to focus on generating Mathematica notebooks on material covered in *Chapter 9 Angular Momentum, Part II (General $\vec{L}$),* which deals the 3-D rigid body dynamics. This topic is notorious difficult/confusing for students and I felt I just skimmed by as a graduate student. I return in retirement after all these years to pay my dues and really understand how to solve problems is this area.

Although Morin’s Chapter 9 guides these notebooks, I made frequent use of other sources such as textbooks available in libraries or on the web. I also found lots of video lectures available on YouTube. Confused about some physics topics? Google it and you’ll be amazed what you find. Some relevant texts are

Classical Mechanics, Hebert Goldstein (my original text at University, late 60s). Newer versions exist.
Mechanics: Volume1 A Course in Theoretical Physics, L. D. Landau and E. M. Lifshitz.
Classical Mechanics, John. R. Taylor
Classical Mechanics of Particles and Systems, Stephen T. Thornton, Jerry B. Marion
Analytical Mechanics, G. R. Fowles, G. L. Cassiday

Analytical Mechanics, Louis N. Hand, Janet D. Finch

I find Mathematica useful for this topic. It facilitates calculations, provides a vehicle for creating instructive visualizations and allows one to quickly generate numerical solutions. Mathematica is a favorite tool of mine but I think it is crucially important to also work with pen and paper. Our brains are closely linked to our hands and one thinks differently with a pen in hand than when sitting before a computer screen. For serious thoughts on this, read *The Craftsman*, by Richard Sennett.

Purpose

I develop some topics related to the free symmetric top, sometimes referred to as the Euler top. The specific topics covered are

1 Precession of free symmetric top in body frame
2 Precession of free symmetric top in lab frame
3 Precession of free symmetric top in lab frame - alternative derivation using Euler angles
4 Space and Body Cones
5 Wobble in the lab frame - numerics

1 Precession of free symmetric top in body frame

Euler’s equations of motion for a rigid body, expressed in terms of principal components, are

```
w1[1] = { I1 D[ω1[t], t] + (I3 - I2) ω2[t] ω3[t]  == τ1,
   I2 D[ω2[t], t] + (I1 - I3) ω1[t] ω3[t]  == τ2,
   I3 D[ω3[t], t] + (I1 - I2) ω1[t] ω2[t]  == τ3}
```

$$\{(-\mathcal{I}_2+\mathcal{I}_3)\,\omega_2[t]\,\omega_3[t]+\mathcal{I}_1\,\omega_1'[t]==\tau_1,\ (\mathcal{I}_1-\mathcal{I}_3)\,\omega_1[t]\,\omega_3[t]+\mathcal{I}_2\,\omega_2'[t]==\tau_2,\ (\mathcal{I}_1-\mathcal{I}_2)\,\omega_1[t]\,\omega_2[t]+\mathcal{I}_3\,\omega_3'[t]==\tau_3\}$$

The term *free* in free symmetric top means there are no applied torques, and the term *symmetric* means that two of the moments of inertia are equal.

```
w1[2] = w1[1] /. τa_ → 0 /. I2 → I1;
w1[2] = Solve[w1[2], {ω1′[t], ω2′[t], ω3′[t]}][[1]] // RE
```

$$\left\{\omega_1'[t]==\frac{(\mathcal{I}_1-\mathcal{I}_3)\,\omega_2[t]\,\omega_3[t]}{\mathcal{I}_1},\ \omega_2'[t]==-\frac{(\mathcal{I}_1-\mathcal{I}_3)\,\omega_1[t]\,\omega_3[t]}{\mathcal{I}_1},\ \omega_3'[t]==0\right\}$$

```
(TraditionalForm /@ w1[2]) // ColumnForm
```

$$\omega_1'(t) = \frac{(I_1 - I_3)\,\omega_2(t)\,\omega_3(t)}{I_1}$$
$$\omega_2'(t) = -\frac{(I_1 - I_3)\,\omega_1(t)\,\omega_3(t)}{I_1}$$
$$\omega_3'(t) = 0$$

These are easily solved. To be consistent with the Figures below, I choose the initial conditions $\vec{\omega}(0) = \omega_0\ e_2 + \omega_{30}\ e_3$

```
w1[3] =
 DSolve[{w1[2], ω1[0] == 0 , ω2[0] == ω0, ω3[0] == ω30}, {ω1[t], ω2[t], ω3[t]}, t][[1]]
```

$$\left\{\omega_3[t] \to \omega_{30},\ \omega_1[t] \to \text{Sin}\left[t\left(\omega_{30} - \frac{I_3\,\omega_{30}}{I_1}\right)\right]\omega_0,\ \omega_2[t] \to \text{Cos}\left[t\left(\omega_{30} - \frac{I_3\,\omega_{30}}{I_1}\right)\right]\omega_0\right\}$$

It is convenient to define

```
def[Ω] = Ω == (I3 - I1)/I1 ω30
```

$$\Omega == \frac{(-I_1 + I_3)\,\omega_{30}}{I_1}$$

and, for later use,

```
def[κ] = κ == I3/I1
```

$$\kappa == \frac{I_3}{I_1}$$

```
w1[4] = w1[3] /. Sol[def[Ω], I1] // Simplify
```

$$\{\omega_3[t] \to \omega_{30},\ \omega_1[t] \to -\text{Sin}[t\,\Omega]\,\omega_0,\ \omega_2[t] \to \text{Cos}[t\,\Omega]\,\omega_0\}$$

The following animation illustrates that the angular frequency $\vec{\omega}$ precesses about the spin axis of the rotating top (e_3 in this case). The direction of precession is positive (counterclockwise) for an oblate top, and negative (clockwise) for a prolate top.

In[81]:=

```
Clear[GenerateFrame];
GenerateFrame[t_] :=
 Module[{ω0 = 0.2, ω30 = 1.0, a1 = 0.5, a2 = 2.0, b = 1, G, frame},
  G[1] = ShowEllipsoidBodyFrame[t, a1, b, ω0, ω30];
  G[2] = ShowEllipsoidBodyFrame[t, a2, b, ω0, ω30];
  frame = Grid[{{G[1], G[2]}}]]

Module[{tMax = 3, frames},
 frames = Table[GenerateFrame[t], {t, 0, tMax, 1}];
 ListAnimate[frames]]
```

Out[83]=



Figure 1 Precession of prolate and oblate ellipsoids in body frame

In the Figure α is the angle between $\vec{\omega}$ and e_3

$$\cos(\alpha) = \frac{\omega_{30}}{\sqrt{2\,\omega_0^2 + \omega_{30}^2}}$$

1 Code

In[65]:=

```
Clear[ShowEllipsoidBodyFrame];
ShowEllipsoidBodyFrame[t_, a_, b_, ω0_, ω30_] :=
 Module[{scale = 2.5, δ = 0.1, vp = {2.5, 1.0, 1}, sz = 0.01, szArrow = 0.03, O,
   e1, e2, e3, I1, I3, Ω, ω1, ω2, ω3, α, αArc, ωVec, ellipsoid, bodyAxes,
   objects, EM, range, latlongcurves, RM = RotationMatrix, lab, G},
  {I1, I3} = {1/5 (a^2 + b^2), 2/5 a^2} // N; (*see Appendix for derivation *)
  Ω = (I3 - I1) ω30 / I1;
  {ω1, ω2, ω3} = {-Sin[t Ω] ω0, Cos[t Ω] ω0, ω30};
  α = ArcCos[ω30 / Sqrt[2 ω0^2 + ω30^2]];

  range = scale {{- 1, 1}, {- 1, 1}, {- 1, 1}};
  {O, e1, e2, e3} = {{0, 0, 0}, {1, 0, 0}, {0, 1, 0}, {0, 0, 1}};

  latlongcurves =
   {LightGray, Table[Line@Table[{a Sin[θθ] Cos[ϕϕ], a Sin[θθ] Sin[ϕϕ], b Cos[θθ] },
       {θθ, 0, π, π / 100}], {ϕϕ, 0, 2 π, π / 12}],
    Table[Line@Table[{a Sin[θθ] Cos[ϕϕ], a Sin[θθ] Sin[ϕϕ], b Cos[θθ] },
       {ϕϕ, 0, 2 π, π / 100}], {θθ, -π, π, π / 8}]};
  bodyAxes = {Blue, Tube[1.3 a {-e1, e1}], Tube[1.3 a { -e2, e2}], Tube[1.3 b {-e3, e3}],
    Tex["x1", 1.5 a e1], Tex["x2", 1.5 a e2], Tex["x3", 1.5 b e3] };
  ellipsoid = {{Opacity[0.25, Blue], Ellipsoid[O, {a, a, b}]},
    latlongcurves, bodyAxes};
  ωVec = {Red, Arrow@Tube[{O, 2 {0, ω0, ω30}}], Stl@Text["ω⃗", 2.2 {0, ω0, ω30}] };
  αArc = {Arrowheads[0.01],
    Arrow@Table[1.2 RM[β, e1].e3, {β, 0, -α, -0.01}], Tex["α", 1.1 RM[-α/2, e1].e3]};

  objects = {ellipsoid, ωVec, αArc};
  objects = GeometricTransformation[objects, RotationMatrix[Ω t, e3]];

  lab = Module[{type },
    type = Which[a == b, "Spheroid",
      a < b, "Prolate ellipsoid",
      a > b, "Oblate ellipsoid"];
    Stl@StringForm["`` in body frame\n a = ``,
       b = ``, I1 = `` I3 = `` ϰ = ``\nPrecession Ω = ``",
      type, NF2@a, NF2@b, NF2@I1, NF2@I3, NF2[N[I3 / I1]], NF2[Ω]]];
  Graphics3D[objects, ImageSize → 400, Axes → False, Boxed → False,
   SphericalRegion → True, ViewPoint → vp, PlotLabel → lab, PlotRange → range]]
```

2 Precession of free symmetric top in lab frame

What happens in the lab frame?

Suppose the free symmetric top is tilted with respect to the lab frame (say θ with respect to to z-axis).

```
Module[{ω0 = 0.1, ω30 = 1.0, θ = π / 6, a1 = 0.5, a2 = 2.0, b = 1, t = 0, G, frame},
 G[1] = ShowEllipsoidLabFrame[t, θ, a1, b, ω0, ω30];
 G[2] = ShowEllipsoidLabFrame[t, θ, a2, b, ω0, ω30];
 Grid[{{G[1], G[2]}}]]
```



Figure 2 Ellipsoids tilted by Euler Angle θ

Because there are no applied torques, the angular momentum $\vec{L}$ is constant. I choose to orient it along the z-Axis. The body axis, x_3, will precess about the fixed angular momentum. To determine the precession rate in the lab frame, calculate the time rate of change of the e_3 unit vector

$$\frac{d\,e_3}{\mathrm{dt}} = \vec{\omega} \times e_3$$

where e_3 is a unit vector for the spin axis in the body frame. Since e_3 is fixed in the body frame, its time rate of change is purely associated with rotation about a unit vector in the direction of the angular moment. I write

An expression for $\vec{\omega}$ in terms of $\vec{L}$ is required. This is such a simple calculation, I'll just illustrate the steps in typeset format. For the free symmetric top

$$\vec{\omega} = \omega_1 e_1 + \omega_2 e_2 + \omega_1 e_3$$

$$\vec{L} = I_1 \omega_1 e_1 + I_1 \omega_2 e_2 + I_3 \omega_1 e_3$$

By using the first expression, $\vec{L}$ can be rewritten

$$\vec{L} = I_1 \Omega e_3 + I_1 (\omega_1 e_2 + \omega_1 e_2 + \omega_3 e_3) = I_1 \Omega e_3 + \vec{\omega} I_1$$

and hence the desired relation of $\vec{\omega}$ as a function of $\vec{L}$.

$$\vec{\omega} = \frac{\vec{L}}{I_1} - I_1 \Omega e_3$$

Then

$$\frac{d e_3}{\text{dt}} = \left(\frac{\vec{L}}{I_1} - I_1 \Omega e_3\right) \times e_3 = \left(\frac{\vec{L}}{I_1}\right) \times e_3 \equiv \vec{\omega}_{\text{precession}} \times e_3$$

The magnitude of the precession angular frequency in the lab frame is

$$\omega_{\text{precession}} = \frac{L}{I_1}$$

2 Code

In[67]:=

```
Clear[ShowEllipsoidLabFrame];
ShowEllipsoidLabFrame[t_, θ_, a_, b_, ω0_, ω30_] :=
 Module[{scale = 2.5, δ = 0.1, vp = {2.5, 1.0, 1}, sz = 0.01, szArrow = 0.03, O, ex,
   ey, ez, labAxes, e1, e2, e3, θArc, I1, I3, Ω, ω1, ω2, ω3, ωVec, LVec, ellipsoid,
   bodyAxes, objects, EM, range, latlongcurves, RM = RotationMatrix, lab, G},
  {I1, I3} = {1/5 (a^2 + b^2), 2/5 a^2} // N; (*see Appendix for derivation *)
  Ω = (I3 - I1) ω30 / I1;
  {ω1, ω2, ω3} = {-Sin[t Ω] ω0, Cos[t Ω] ω0, ω30};

  range = scale {{- 1, 1}, {- 1, 1}, {- 1, 1}};
  {O, ex, ey, ez} = {{0, 0, 0}, {1, 0, 0}, {0, 1, 0}, {0, 0, 1}};
  labAxes = {Black, Line[1.3 {-ex, ex}], Line[1.3 { -ey, ey}],
    Line[1.3 {-ez, ez}], Tex["x", 1.5 ex], Tex["y", 1.5 ey], Tex["z", 1.9 ez ] };
  {e1, e2, e3} = {{1, 0, 0}, {0, 1, 0}, {0, 0, 1}};
  θArc =
   {Arrowheads[0.01], Arrow@Table[0.75 RM[α, e1].ez, {α, 0, Max[θ, 0.01], 0.05}],
    Tex["θ", 0.65 RM[θ/2, e1].ez]};

  latlongcurves =
   {LightGray, Table[Line@Table[{a Sin[θθ] Cos[ϕϕ], a Sin[θθ] Sin[ϕϕ], b Cos[θθ] },
       {θθ, 0, π, π/100}], {ϕϕ, 0, 2 π, π/12}],
    Table[Line@Table[{a Sin[θθ] Cos[ϕϕ], a Sin[θθ] Sin[ϕϕ], b Cos[θθ] },
       {ϕϕ, 0, 2 π, π/100}], {θθ, -π, π, π/8}]};
  bodyAxes = {Blue, Tube[1.3 a {-e1, e1}], Tube[1.3 a { -e2, e2}], Tube[1.3 b {-e3, e3}],
    Tex["x1", 1.5 a e1], Tex["x2", 1.5 a e2], Tex["x3", 1.5 b e3] };
  ellipsoid = {{Opacity[0.15, Blue], Ellipsoid[O, {a, a, b}]},
    latlongcurves, bodyAxes};
  ωVec = With[{vScale = 1.5}, {Red, Arrow@Tube[{O, vScale {0, ω0, ω30}}],
     Stl@Text["ω⃗", 1.1 vScale {0, ω0, ω30}] }];
  LVec = With[{vScale = 1.5}, {Darker[Green, 0.5],
     Arrow@Tube[{O, vScale ez}], Stl@Text["L⃗", 1.1 vScale ez] }];
  objects = {ellipsoid, ωVec};
  objects = GeometricTransformation[objects, RotationMatrix[θ, e1]];

  lab = Module[{type },
    type = Which[a == b, "Spheroid",
      a < b, "Prolate ellipsoid",
      a > b, "Oblate ellipsoid"];
    Stl@StringForm["`` in lab frame\n a = ``, b = ``, I1 = `` I3 = `` κ = ``",
      type, NF2@a, NF2@b, NF2@I1, NF2@I3, NF2[N[I3/I1]]]];
  Graphics3D[{labAxes, LVec, θArc, objects}, ImageSize → 400,
   Axes → False, Boxed → False, SphericalRegion → True,
   ViewPoint → vp, PlotLabel → lab, PlotRange → range]]
```

3 Precession of free symmetric top in lab frame - alternative derivation using Euler angles

I perform a second derivation of the precession frequency in the lab frame. This one uses Euler angles (see notebook *Euler's Equations and Euler Angles 11-08-18*) and is more informative about the geometry. First, I illustrate some relationships that exist because $\vec{L}$, $\vec{\omega}$ and e_3.

```
Module[{θ = π / 6, a1 = 1.0, b1 = 1.5, a2 = 2.0, b2 = 1.0, G},
 G[1] = ShowPlanarDiagram[θ, a1, b1];
 G[2] = ShowPlanarDiagram[θ, a2, b2];
 Grid[{{G[1], G[2]}}]]
```



Figure 3 Relationship of $\vec{L}$, $\vec{\omega}$ and $\vec{e}_3$

From the diagram,

```
w3[1] = {L2 == L Sin[θ], L3 == L Cos[θ]}
```

```
{L2 == L Sin[θ], L3 == L Cos[θ]}
```

```
w3[2] = {ω2 == ω Sin[α], ω3 == ω Cos[α]}
```

```
{ω2 == ω Sin[α], ω3 == ω Cos[α]}
```

Also, from $\vec{L} = \overleftrightarrow{\mathcal{I}} \cdot \vec{\omega}$,

```
w3[3] = w3[1] /. {L2 → I1 ω2, L3 → I3 ω3} /. (w3[2] // ER)
```

$\{\omega \operatorname{Sin}[\alpha]\, \mathcal{I}_1 == \operatorname{L} \operatorname{Sin}[\theta],\ \omega \operatorname{Cos}[\alpha]\, \mathcal{I}_3 == \operatorname{L} \operatorname{Cos}[\theta]\}$

```
w3[4] = Eliminate[w3[3], L]
```

$\omega \operatorname{Cos}[\alpha] \operatorname{Sin}[\theta]\, \mathcal{I}_3 == \omega \operatorname{Cos}[\theta] \operatorname{Sin}[\alpha]\, \mathcal{I}_1$

An equation relating the angles θ and α to the moments of inertia is obtained.

```
w3[5] = MapEqn[(#/(ω Cos[θ] Cos[α])) &, w3[4]]
```

$\mathcal{I}_3 \operatorname{Tan}[\theta] == \mathcal{I}_1 \operatorname{Tan}[\alpha]$

From Notebook *Euler's Equations and Euler Angles 11-08-18*

Euler Angles

$$\frac{d\vec{\phi}}{dt} = \dot{\phi}\,\hat{e}_z,\ \frac{d\vec{\theta}}{dt} = \dot{\theta}\,\hat{e}_1,\ \frac{d\vec{\psi}}{dt} = \dot{\psi}\,\hat{e}_3$$



Figure 4 Time dependencies

In terms of Euler angles, the angular frequency is

$$\vec{\omega} = \vec{\omega} = \omega_1 e_1 + \omega_2 e_2 + \omega_3 e_3 = \dot{\vec{\phi}} + \dot{\vec{\theta}} + \dot{\vec{\psi}}$$

Specifically, it was shown that

```
w3[6] =
  {ω1 == Cos[ψ] θ̇ + ϕ̇ Sin[θ] Sin[ψ], ω2 == Cos[ψ] ϕ̇ Sin[θ] - θ̇ Sin[ψ], ω3 == Cos[θ] ϕ̇ + ψ̇}

{ω1 == Cos[ψ] θ̇ + ϕ̇ Sin[θ] Sin[ψ], ω2 == Cos[ψ] ϕ̇ Sin[θ] - θ̇ Sin[ψ], ω3 == Cos[θ] ϕ̇ + ψ̇}
```

It can be seen that the precession rate about the e_z axis may also be expressed as

$$\omega_{\text{precession}} = \frac{d\phi}{dt}$$

Consideration of the $\omega 2$ component from expressions about provides an equation for $d\phi/dt$. From purely geometric considerations

```
w3[7] = w3[2][[1]]

ω2 == ω Sin[α]
```

From Euler angle relationships

```
w3[8] = w3[6][[2]]

ω2 == Cos[ψ] ϕ̇ Sin[θ] - θ̇ Sin[ψ]
```

```
w3[9] = w3[7][[2]] == w3[8][[2]]

ω Sin[α] == Cos[ψ] ϕ̇ Sin[θ] - θ̇ Sin[ψ]
```

The plane depicted in Figure 1 corresponds to $\psi = 0$.

```
w3[10] = w3[9] /. ψ → 0

ω Sin[α] == ϕ̇ Sin[θ]
```

This can be further simplified by removing θ with the aid of

```
w3[5]

I3 Tan[θ] == I1 Tan[α]
```

An expression for Sin[θ] is easily obtained

```
sideCalc[1] = Solve[w3[5] /. Tan[θ] → S/Sqrt[1 - S^2], S][[2, 1]] /. S → Sin[θ]
```

$$\text{Sin}[\theta] \to \frac{\mathcal{I}_1 \,\text{Tan}[\alpha]}{\sqrt{\mathcal{I}_3^2 + \mathcal{I}_1^2 \,\text{Tan}[\alpha]^2}}$$

Thus

```
w3[11] = w3[10] /. sideCalc[1]
```

$$\omega \, \text{Sin}[\alpha] == \frac{\dot{\phi} \, \mathcal{I}_1 \, \text{Tan}[\alpha]}{\sqrt{\mathcal{I}_3^2 + \mathcal{I}_1^2 \, \text{Tan}[\alpha]^2}}$$

and

```
w3[12] = Sol[w3[11], ϕ̇]
```

$$\dot{\phi} \rightarrow \frac{\omega \, \text{Cos}[\alpha] \, \sqrt{\mathcal{I}_3^2 + \mathcal{I}_1^2 \, \text{Tan}[\alpha]^2}}{\mathcal{I}_1}$$

This can be further simplified to a form found in the literature by some rather arcane Mathematica operations

```
w3[13] = Simplify[w3[12] /. Sol[def[κ], I3], Assumptions → {I1 > 0}]
```

$$\dot{\phi} \rightarrow \omega \, \text{Cos}[\alpha] \, \sqrt{\kappa^2 + \text{Tan}[\alpha]^2}$$

```
w3[14] = w3[13] /. a_ Cos[α] Sqrt[b_] :> a Sqrt[Expand[Cos[α]^2 b]]
```

$$\dot{\phi} \rightarrow \omega \, \sqrt{\kappa^2 \, \text{Cos}[\alpha]^2 + \text{Sin}[\alpha]^2}$$

```
w3[15] = w3[14] /. Sin[α]^2 → 1 - Cos[α]^2 /. a_ Sqrt[b_] :> a Sqrt[Collect[b, {Cos[α]^2}]]
```

$$\dot{\phi} \rightarrow \omega \, \sqrt{1 + \left(-1 + \kappa^2\right) \, \text{Cos}[\alpha]^2}$$

3 Code

```
Clear[ShowPlanarDiagram];
ShowPlanarDiagram[θ_, a_, b_] :=
 Module[{scale = 1, δ = 0.1, O, ey, ez, e2, e3, e3Vec,
   ω, α, I1, I3, κ, axesyz, axes23, RM = RotationMatrix, Stl2, Tex},
  Stl2[x_] := Style[x, Bold, FontFamily → "Helvetica", 10];
  Tex[text_, position_] := Text[Style[text, Bold, FontSize → 10], position];
  {O, ey, ez} = {{0, 0}, {1, 0}, {0, 1}};
  axesyz = {Black, Line[scale { -ey, ey}],
    Line[scale {- ez, ez}], Tex["y", (scale + δ) ey], Tex["z", (scale + δ) ez]};
  (* RM[θ].ex  where . means "keyboard period" is short
   hand for Dot[RotationMatrix[θ], ex] *)
  {e2, e3} = (RM[θ].#) & /@ {ey, ez};
  axes23 = {Blue, Line[scale { -e2, e2}], Line[scale {- e3, e3}],
    Tex["x₂", (scale + δ) e2], Tex["x₃", (scale + δ) e3]};
  {I1, I3} = {1/5 (a² + b²), 2/5 a²} // N; (*see Appendix for derivation *)
  κ = I3 / I1;
  α = ArcTan[κ Tan[θ]];

  Module[{ωVec, LVec, θArc, αArc, lab},
   ωVec = With[{r = 0.5, ptω = RM[-α].e3},
     {Red, Arrow[{O, r ptω}], Tex["ω⃗", 1.1 r ptω]}];
   e3Vec = With[{r = 0.25},
     {Directive[Black, Thick], Arrow[{O, r e3}], Tex["e₃", 1.15 r e3]}];
   LVec = With[{LMag = 0.7},
     {Darker[Green, 0.5], Arrow[{O, {0, LMag}}], Tex["L⃗", {0, 1.1 LMag}] }];
   θArc = With[{r = 0.9, θS = 0, θF = θ, text = "θ"},
     {Arrowheads[Small], Arrow@Table[r RM[ξ].{0, 1}, {ξ, θS, θF, (θF - θS)/20}],
      Text["θ", 0.9 r RM[θ / 2].{0, 1}]}];
   αArc = With[{r = 0.4, θS = θ, θF = θ - α, text = "α"},
     {Arrowheads[Small], Arrow@Table[r Dot[RM[ξ], {0, 1}], {ξ, θS, θF, (θF - θS)/20}],
      Text["α", 0.8 r Dot[RM[θS - α/2], {0, 1}]]}];

   lab = Module[{type },
     type = Which[a == b, "Spheroid",
       a < b, "Prolate ellipsoid",
       a > b, "Oblate ellipsoid"];
     Stl2@StringForm["Coplanar relationship of L⃗,
        ω⃗, e₃\n`` a = ``, b = ``, I₁ = `` I₃ = `` κ = ``",
       type, NF2@a, NF2@b, NF2@I1, NF2@I3, NF2[N[I3 / I1]]]];
   Graphics[{axesyz, axes23, θArc, αArc, ωVec, LVec, e3Vec},
    PlotLabel → lab, ImageSize → 300]]]
```

4 Space and Body Cones

The precession of free tops has an interesting history of being described in terms of geometrical forms. For example, many features of the motion of free asymmetric tops can be described using Poinsot's construction, geometrical ellipsoidal forms based on the conservation of energy and angular momentum. The focus in this notebook is on free symmetric tops and the precession can be visualized using rotating "space" and "body" cones.

```
Module[{θ = π / 6, a1 = 0.5, a2 = 1.5, b = 1.0, G},
 G[1] = ShowSpaceBodyCones[θ, a1, b];
 G[2] = ShowSpaceBodyCones[θ, a2, b];
 Grid[{{G[1], G[2]}}]]
```



Figure 5 Space and body cones

The red cone is the body cone, centered about the spin axis (the x_3 body coordinate, which is rotated by Euler angle θ from the lab frame z-axis. The green cone is the space cone, centered about the constant angular momentum $\vec{L}$ which is oriented in the lab frame z-axis. The two-dimensional projection of these cones onto the x_2 - x_3 plane corresponds to the view shown in Figure 3. That Figure can be consulted for information about the particular relevant angles and vectors. In the case of a prolate top, the body cone rotates on the outside surface of the space cone while, for an oblate top, the space cone is inside the body cone.

The two precessional motions, precession of $\vec{\omega}$ about the axis x_3 at speed Ω and precession of the axis of the top about the angular moment at speed $\dot{\phi}$ is illustrated in the following animations. Note the opposite directions of the two precessional motions for the prolate case.

In[86]:=
```
Module[{tMax = 3, θ = π / 6, a1 = 0.5, a2 = 1.5, b = 1.0, ω = 1, frames},
 frames  = Table[ShowSpaceBodyConesFrame[t, θ, a1, b, ω], {t, 0, tMax, 0.5}];
 ListAnimate[frames]]
```

Out[86]=



In[87]:=

```
Module[{tMax = 2, θ = π / 6, a1 = 0.5, a2 = 1.5, b = 1.0, ω = 1, frames},
 frames = Table[ShowSpaceBodyConesFrame[t, θ, a2, b, ω], {t, 0, tMax, 0.25}];
 ListAnimate[frames]]
```

Out[87]=



4 Code

In[84]:=

```
Clear[ShowSpaceBodyConesFrame];
ShowSpaceBodyConesFrame[t_, θ_, a_, b_, ω_] :=
 Module[{scale = 1.1, δ = 0.1, vp = {2.5, 1.0, 1},
   sz = 0.01, szArrow = 0.03, range, O, ex, ey, ez, axesLab, e1, e2,
   e3, axesBody, 𝓘1, 𝓘3, κ, α, β, lBoundary, RM = RotationMatrix,
   coneBody, coneSpace, objects, ϕPrecession, ΩPrecession, Stl2},
  Stl2[x_] := Style[x, Bold, FontFamily → "Helvetica", 10];
  range = scale {{-1, 1}, {-1, 1}, {-1, 1}};
  {O, ex, ey, ez} = {{0, 0, 0}, {1, 0, 0}, {0, 1, 0}, {0, 0, 1}};
  axesLab =
   {Black, Line[scale {-ex, ex}], Line[scale { -ey, ey}], Line[scale {-0.1 ez, ez}],
    Tex["x", 1.1 scale ex], Tex["y", 1.1 scale ey], Tex["z", 1.1 scale ez ] };
```

```
{e1, e2, e3} = (RM[θ, ex].#) & /@ {ex, ey, ez};
axesBody = With[{scale123 = 1.3 scale},
  {Black, Line[scale123 {-e1, e1}], Line[scale123 { -e2, e2}],
   Line[scale123 {-0.1 e3, e3}], Tex["x1", 1.1 scale123 e1],
   Tex["x2", 1.1 scale123 e2], Tex["x3", 1.1 scale123 e3 ] }];

{I1, I3} = {1/5 (a^2 + b^2), 2/5 a^2} // N;
(* Symmetric ellipsoid, see Appendix for derivation *)
κ = I3/I1;
α = ArcTan[κ Tan[θ]];
ΩPrecession  =  (κ - 1) ω Cos[α];
ϕPrecession = ω Sqrt[1 + (κ^2 - 1) Cos[α]^2];

lBoundary = 1; (* Length of common boundary line of cones *)
β = If[κ < 1, θ - α, α - θ];
coneSpace = With[{rSpace = lBoundary Sin[β], lSpace = lBoundary Cos[β]},
  {{Opacity[0.25, Green], Cone[{lSpace ez, O}, rSpace]},
   {Darker[Green, 0.5], Line[{O, {0, -rSpace, lSpace}}],
    PointSize[0.02], Point[{0, -rSpace, lSpace}] }}];

(* Create body cone structure in lab frame *)
coneBody = With[{rBody = lBoundary Sin[α], lBody = lBoundary Cos[α]},
  {{Opacity[0.25, Red], Cone[{lBody ez, O}, rBody]},
   {Darker[Red, 0.5], Line[{O, {0, rBody, lBody}}],
    PointSize[0.02], Point[{0, rBody, lBody}] }}];
(* Rotate body cone structure to align with e3 axis *)
coneBody = GeometricTransformation[coneBody, RotationMatrix[θ, ex]];
(* Precess the body cone by Ω t *)
coneBody = GeometricTransformation[coneBody, RotationMatrix[ΩPrecession t, e3]];

objects = {coneSpace, coneBody};
(* Precess objects by ϕ̇ t *)
objects = GeometricTransformation[objects, RotationMatrix[ϕPrecession t, ez]];

Module[{lab},
 lab = Module[{type },
   type = Which[a == b, "Spheroid",
     a < b, "Prolate ellipsoid",
     a > b, "Oblate ellipsoid"];
   Stl2@
    StringForm["Space (green) and body (red) cones for ``\na = ``, b = ``, I1 =
       `` I3 = `` κ = ``\nΩ = ``  ϕ̇ = ``", type, NF2@a, NF2@b, NF2@I1,
     NF2@I3, NF2[N[I3/I1]], NF2@ΩPrecession, NF2@ϕPrecession]];
 Graphics3D[{axesLab, objects}, ImageSize → 400, Axes → False,
  Boxed → False, SphericalRegion → True, ViewPoint → vp,
  ViewVertical → {0, 0, 1}, PlotLabel → lab, PlotRange → range]]
```

```
Clear[ShowSpaceBodyCones];
ShowSpaceBodyCones[θ_, a_, b_] :=
 Module[{scale = 1.1, δ = 0.1, vp = {2.5, 1.0, 1}, sz = 0.01, szArrow = 0.03, range,
   O, ex, ey, ez, axesLab, e1, e2, e3, axesBody, I1, I3, κ, α, β, lBoundary,
   RM = RotationMatrix, coneSymmetry, coneBody, coneSpace, objectsBody, Stl2},
  Stl2[x_] := Style[x, Bold, FontFamily → "Helvetica", 10];
  range =  scale {{- 1, 1}, {- 1, 1}, {- 1,  1}};
  {O, ex, ey, ez} = {{0, 0, 0}, {1, 0, 0}, {0, 1, 0}, {0, 0, 1}};
  axesLab =
   {Black, Line[scale {-ex, ex}], Line[scale { -ey, ey}], Line[scale {-0.1 ez, ez}],
    Tex["x", 1.1 scale ex], Tex["y", 1.1 scale ey], Tex["z", 1.1 scale ez ] };
  {e1, e2, e3} = (RM[θ, ex].#) & /@ {ex, ey, ez};
  axesBody = With[{scale123 = 1.3 scale},
    {Black, Line[scale123 {-e1, e1}], Line[scale123 { -e2, e2}],
     Line[scale123 {-0.1 e3, e3}], Tex["x1", 1.1 scale123 e1],
     Tex["x2", 1.1 scale123 e2], Tex["x3", 1.1 scale123 e3 ] }];

  {I1, I3} = {1/5 (a^2 + b^2), 2/5 a^2} // N;
  (* Symmetric ellipsoid, see Appendix for derivation *)
  κ = I3 / I1;
  α = ArcTan[κ Tan[θ]];

  lBoundary = 1; (* Length of common boundary line of cones *)
  coneSymmetry = {Opacity[0.25, Blue], Cone[1.25 {ez, O}, Cos[θ]]};
  coneSymmetry = With[{rSym = 1.25 Tan[θ], lSym = 1.25 lBoundary},
    {Opacity[0.25, Blue], Cone[{lSym ez, O}, rSym]}];
  β = If[κ < 1, θ - α, α - θ];
  coneSpace = With[{rSpace = lBoundary Sin[β], lSpace = lBoundary Cos[β]},
    {Opacity[0.25, Green], Cone[{lSpace ez, O}, rSpace]}];
  coneBody = With[{rBody = lBoundary Sin[α], lBody = lBoundary Cos[α]},
    {Opacity[0.25, Red], Cone[{lBody e3, O}, rBody]}];
  objectsBody = {Black,
    GeometricTransformation[{Line[1.5 {O, e3}], coneBody}, RotationMatrix[θ, e1]]};

  Module[{lab},
   lab = Module[{type },
     type = Which[a == b, "Spheroid",
       a < b, "Prolate ellipsoid",
       a > b, "Oblate ellipsoid"];
     Stl2@StringForm["Space (green) and body (red)
         cones for ``\na = ``, b = ``, I1 = `` I3 = `` κ = ``",
       type, NF2@a, NF2@b, NF2@I1, NF2@I3, NF2[N[I3 / I1]]]];
   Graphics3D[{axesLab, axesBody, (*coneSymmetry,*) coneSpace, coneBody},
    ImageSize → 400, Axes → False, Boxed → False, SphericalRegion → True,
    ViewPoint → vp, PlotLabel → lab, PlotRange → range]]]
```

5 Wobble in the lab frame - numerics

A free symmetric top is said to precess in the body frame but wobble in the lab frame. However, to see the wobble, one needs the time dependence of the Euler angle representation of the top. Obtaining this requires a numerical solution of the equation of motion. I start with the Euler equations specialized to a free symmetric top (equation w1[2] above) and prepare those equations for numerical integration.

```
w5[1] = {ω1'[t] == ((I1 - I3) ω2[t] ω3[t])/I1, ω2'[t] == -((I1 - I3) ω1[t] ω3[t])/I1, ω3'[t] == 0}
```

$$\left\{\omega_1'[t] == \frac{(I_1 - I_3)\,\omega_2[t]\,\omega_3[t]}{I_1},\ \omega_2'[t] == -\frac{(I_1 - I_3)\,\omega_1[t]\,\omega_3[t]}{I_1},\ \omega_3'[t] == 0\right\}$$

The third equation implies $\omega_3 = \omega_{30}$ is constant. I make use of the parameter κ

```
w5[2] = w5[1][[1 ;; 2]] /. ω3[t] → ω30 /. I3 → κ I1 // Expand
```

$$\{\omega_1'[t] == \omega_{30}\,\omega_2[t] - \kappa\,\omega_{30}\,\omega_2[t],\ \omega_2'[t] == -\omega_{30}\,\omega_1[t] + \kappa\,\omega_{30}\,\omega_1[t]\}$$

I introduce dimensionless variables

```
w5[3] = w5[2] /.
    {ω1 → Function[{t}, Ω1[ω30 t]], ω2 → Function[{t}, Ω2[ω30 t]]} /. t → T / ω30;
w5[3] = MapEqn[(Simplify[# / ω30]) &, w5[3]]
```

$$\{\Omega1'[T] == -(-1 + \kappa)\,\Omega2[T],\ \Omega2'[T] == (-1 + \kappa)\,\Omega1[T]\}$$

These equation are, of course, trivial to integrate in the body frame. However, I require the solution in the lab frame and must use the Euler angles. From Euler's Equations and Euler Angles 11-08-18 the rotation frequencies in lab coordinates are

```
w5[6] =
 {ω1 == Cos[ϕ] θ̇ + ψ̇ Sin[θ] Sin[ϕ], ω2 == -Cos[ϕ] ψ̇ Sin[θ] + θ̇ Sin[ϕ], ω3 == ϕ̇ + Cos[θ] ψ̇}
```

$$\{\omega_1 == \mathrm{Cos}[\phi]\,\dot\theta + \dot\psi\,\mathrm{Sin}[\theta]\,\mathrm{Sin}[\phi],\ \omega_2 == -\mathrm{Cos}[\phi]\,\dot\psi\,\mathrm{Sin}[\theta] + \dot\theta\,\mathrm{Sin}[\phi],\ \omega_3 == \dot\phi + \mathrm{Cos}[\theta]\,\dot\psi\}$$

Make the notation consistent with that used in this notebook

```
w5[7] = w5[6] /. {ω1 → ω1[t], ω2 → ω2[t], ω3 → ω30, ϕ̇ → D[ϕ[t], t],
   θ̇ → D[θ[t], t], ψ̇ → D[ψ[t], t], ϕ → ϕ[t], θ → θ[t], ψ → ψ[t]}
```

$$\{\omega_1[t] == \mathrm{Cos}[\phi[t]]\,\theta'[t] + \mathrm{Sin}[\theta[t]]\,\mathrm{Sin}[\phi[t]]\,\psi'[t],$$
$$\omega_2[t] == \mathrm{Sin}[\phi[t]]\,\theta'[t] - \mathrm{Cos}[\phi[t]]\,\mathrm{Sin}[\theta[t]]\,\psi'[t],\ \omega_{30} == \phi'[t] + \mathrm{Cos}[\theta[t]]\,\psi'[t]\}$$

Choose the same dimensionless variables

```
w5[8] = w5[7] /. ω3[t] → ω30 /. {ω1 → Function[{t}, ω30 Ω1[ω30 t]],
     ω2 → Function[{t}, ω30 Ω2[ω30 t]], ϕ → Function[{t}, ϕ[ω30 t]],
     θ → Function[{t}, θ[ω30 t]], ψ → Function[{t}, ψ[ω30 t]]} /. t → T / ω30;
w5[8] = MapEqn[(Simplify[# / ω30]) &, w5[8]]

{Ω1[T] == Cos[ϕ[T]] θ′[T] + Sin[θ[T]] Sin[ϕ[T]] ψ′[T],
 Ω2[T] == Sin[ϕ[T]] θ′[T] - Cos[ϕ[T]] Sin[θ[T]] ψ′[T], 1 == ϕ′[T] + Cos[θ[T]] ψ′[T]}
```

Next, for the free symmetric top, the angle θ is constant

```
w5[9] = w5[8] /. θ → Function[{T}, θ0]

{Ω1[T] == Sin[θ0] Sin[ϕ[T]] ψ′[T],
 Ω2[T] == -Cos[ϕ[T]] Sin[θ0] ψ′[T], 1 == ϕ′[T] + Cos[θ0] ψ′[T]}
```

The first two equation provide explicit expressions for ϕ′[T] and ψ′[T]

```
w5[10] = w5[9][[1 ;; 2]]

{Ω1[T] == Sin[θ0] Sin[ϕ[T]] ψ′[T], Ω2[T] == -Cos[ϕ[T]] Sin[θ0] ψ′[T]}
```

Derivatives of these equations are also required

```
w5[11] = MapEqn[D[#, T] &, w5[10]]

{Ω1′[T] == Cos[ϕ[T]] Sin[θ0] ϕ′[T] ψ′[T] + Sin[θ0] Sin[ϕ[T]] ψ′′[T],
 Ω2′[T] == Sin[θ0] Sin[ϕ[T]] ϕ′[T] ψ′[T] - Cos[ϕ[T]] Sin[θ0] ψ′′[T]}
```

Substitute these last two equations into the dimensionless Euler equations

```
w5[12] = w5[3] /. (w5[10] // ER) /. (w5[11] // ER)

{Cos[ϕ[T]] Sin[θ0] ϕ′[T] ψ′[T] + Sin[θ0] Sin[ϕ[T]] ψ′′[T] ==
   (-1 + κ) Cos[ϕ[T]] Sin[θ0] ψ′[T],
 Sin[θ0] Sin[ϕ[T]] ϕ′[T] ψ′[T] - Cos[ϕ[T]] Sin[θ0] ψ′′[T] ==
   (-1 + κ) Sin[θ0] Sin[ϕ[T]] ψ′[T]}
```

The equations to be numerically solved are

```
EQNS = Join[w5[12], {ϕ[0] == ϕ0, ϕ′[0] == dϕ0, ψ[0] == ψ0, ψ′[0] == dψ0}]

{Cos[ϕ[T]] Sin[θ0] ϕ′[T] ψ′[T] + Sin[θ0] Sin[ϕ[T]] ψ′′[T] ==
   (-1 + κ) Cos[ϕ[T]] Sin[θ0] ψ′[T],
 Sin[θ0] Sin[ϕ[T]] ϕ′[T] ψ′[T] - Cos[ϕ[T]] Sin[θ0] ψ′′[T] ==
   (-1 + κ) Sin[θ0] Sin[ϕ[T]] ψ′[T], ϕ[0] == ϕ0, ϕ′[0] == dϕ0, ψ[0] == ψ0, ψ′[0] == dψ0}
```

```
EQNS // ColumnForm
```

```
Cos[ϕ[T]] Sin[θ0] ϕ′[T] ψ′[T] + Sin[θ0] Sin[ϕ[T]] ψ′′[T] == (-1 + κ) Cos[ϕ[T]] Sin[θ0] ψ′[T]
Sin[θ0] Sin[ϕ[T]] ϕ′[T] ψ′[T] - Cos[ϕ[T]] Sin[θ0] ψ′′[T] == (-1 + κ) Sin[θ0] Sin[ϕ[T]] ψ′[T]
ϕ[0] == ϕ0
ϕ′[0] == dϕ0
ψ[0] == ψ0
ψ′[0] == dψ0
```

For some representative parameters

In[89]:=
```
TMAX = 40;
SOLN =
  Module[{κ = 0.1, θ0 = π / 6, ψ0 = 0, dψdT0 = 1, ϕ0 = 0, dϕdT0 = 0.1, eqns, soln},
   eqns = {Cos[ϕ[T]] Sin[θ0] ϕ′[T] ψ′[T] + Sin[θ0] Sin[ϕ[T]] ψ′′[T] == - (1 - κ) Cos[ϕ[T]]
       Sin[θ0] ψ′[T], Sin[θ0] Sin[ϕ[T]] ϕ′[T] ψ′[T] - Cos[ϕ[T]] Sin[θ0] ψ′′[T] ==
      (-1 + κ) Sin[θ0] Sin[ϕ[T]] ψ′[T], ψ[0] == ψ0, ψ′[0] == dψdT0, ϕ[0] == ϕ0};
   NDSolve[eqns, {ϕ, ψ}, {T, 0, TMAX}]];
```

As could be expected for constant spin and precession, ϕ and ψ increase linearly with time

```
Plot[{ϕ[T] /. SOLN, ψ[T] /. SOLN}, {T, 0, TMAX}, PlotStyle → {Black, Blue},
 AxesLabel → {Stl["T"], ""}, PlotLegends → {Stl["ϕ[T]"], Stl["ψ[T]"]}]
```



In[91]:=

```
Module[{ω0 = 0.1, ω30 = 1.0, θ = π/12,
  a1 = 0.5, a2 = 2.0, b = 1, dT = 0.5, G, frames},
 frames = Table[ShowEllipsoidWobbleLabFrameUsingEulerMatrix[T, θ, a2, b, ω0, ω30],
   {T, 0, TMAX/10, dT}];
 ListAnimate[frames] ]
```

Out[91]=



In[69]:=

```
Clear[ShowEllipsoidWobbleLabFrameUsingEulerMatrix];
ShowEllipsoidWobbleLabFrameUsingEulerMatrix[t_, θ_, a_, b_, ω0_, ω30_] :=
 Module[{scale = 2.5, δ = 0.1, vp = {2.5, 1.0, 1}, sz = 0.01, szArrow = 0.03, O, ex,
   ey, ez, labAxes, e1, e2, e3, I1, I3(*,ω1, ω2, ω3*), ωVec, LVec, ellipsoid,
   bodyAxes, objects, EM, range, latlongcurves, RM = RotationMatrix, lab, G},
  {I1, I3} = {1/5 (a^2 + b^2), 2/5 a^2} // N; (*see Appendix for derivation *)

  range = scale {{-1, 1}, {-1, 1}, {-1, 1}};
  {O, ex, ey, ez} = {{0, 0, 0}, {1, 0, 0}, {0, 1, 0}, {0, 0, 1}};
  labAxes = {Black, Line[1.3 {-ex, ex}], Line[1.3 { -ey, ey}],
    Line[1.3 {-ez, ez}], Tex["x", 1.5 ex], Tex["y", 1.5 ey], Tex["z", 1.9 ez ] };
  {e1, e2, e3} = {{1, 0, 0}, {0, 1, 0}, {0, 0, 1}};

  latlongcurves =
   {LightGray, Table[Line@Table[{a Sin[θθ] Cos[ϕϕ], a Sin[θθ] Sin[ϕϕ], b Cos[θθ] },
       {θθ, 0, π, π/100}], {ϕϕ, 0, 2 π, π/12}],
    Table[Line@Table[{a Sin[θθ] Cos[ϕϕ], a Sin[θθ] Sin[ϕϕ], b Cos[θθ] },
       {ϕϕ, 0, 2 π, π/100}], {θθ, -π, π, π/8}]};
  bodyAxes = {Blue, Tube[1.3 a {-e1, e1}], Tube[1.3 a { -e2, e2}], Tube[1.3 b {-e3, e3}],
    Tex["x1", 1.5 a e1], Tex["x2", 1.5 a e2], Tex["x3", 1.5 b e3] };
  ellipsoid = {{Opacity[0.15, Blue], Ellipsoid[O, {a, a, b}]},
    latlongcurves, bodyAxes};
  ωVec = With[{vScale = 1.5}, {Red, Arrow@Tube[{O, vScale {0, ω0, ω30}}],
     Stl@Text["ω⃗", 1.1 vScale {0, ω0, ω30}] }];
  LVec = With[{vScale = 1.5}, {Darker[Green, 0.5],
     Arrow@Tube[{O, vScale ez}], Stl@Text["L⃗", 1.1 vScale ez] }];
  objects = {ellipsoid, ωVec};

  (* The ellipsoid has been constructed in the body
   frame. It is transformed into the body frame using the Euler
   matrix and the numerically determined values of ϕ and ψ *)
  EM = EulerMatrix[{ϕ[t], θ, ψ[t]}, {3, 1, 3}] /. SOLN;
  objects =
   GeometricTransformation[objects, EM];

  lab = Module[{type },
    type = Which[a == b, "Spheroid",
      a < b, "prolate",
      a > b, "oblate"];
    Stl@StringForm[
      "Wobbling `` ellipsoid in lab frame\n a = ``, b = ``, I1 = `` I3 = `` κ = ``",
      type, NF2@a, NF2@b, NF2@I1, NF2@I3, NF2[N[I3/I1]]]];
  Graphics3D[{labAxes, LVec, objects}, ImageSize → 400, Axes → False, Boxed → False,
   SphericalRegion → True, ViewPoint → vp, PlotLabel → lab, PlotRange → range]]
```

Appendix A Moment of inertia of symmetric ellipsoid

Moment of inertia of symmetric ellipsoid

```
With[{ρ = M/((4/3) π a^2 b)},
  ρ Integrate[
    {{y^2 + z^2, -x y, -x z},
      {-x y, x^2 + z^2, -y z},
      {-x z, -y z, x^2 + y^2}} Boole[(x^2 + y^2)/a^2 + z^2/b^2 < 1],
    {x, -∞, ∞}, {y, -∞, ∞}, {z, -∞, ∞}, Assumptions → a > 0 && b > 0]
 ] // TraditionalForm
```

$$\begin{pmatrix} \frac{1}{5}(a^2+b^2)M & 0 & 0 \\ 0 & \frac{1}{5}(a^2+b^2)M & 0 \\ 0 & 0 & \frac{2a^2M}{5} \end{pmatrix}$$

Graphics Utilities

In[71]:=

```
(* Helper Functionsc*)
Clear[StoC, Tex, Vec, VecLab, Mess, DimMarker3];
StoC[r_, θ_, ϕ_] := {r Sin[θ] Cos[ϕ], r Sin[θ] Sin[ϕ], r Cos[θ] };
Tex[text_, position_] := Text[Style[text, Bold, FontSize → 10], position];
Vec[vec_] := {Arrowheads[0.05], Arrow[Tube[vec, 0.02]]};
Vec[vec_, size_, sizeAH_] := {Arrowheads[sizeAH], Arrow[Tube[vec, size]]};
(* Draw vector with label place beyond the tip *)
VecLab[st_, fn_, sz_, szArrow_, txt_, txtScale_] :=
 Module[{vec, vecLabel},
  vec = Vec[{st, fn}, sz, szArrow];
  vecLabel = Text[Stl[txt], st + txtScale (fn - st)];
  {vec, vecLabel}]
Mess[lab_, Pobj_, offset_] :=
  Module[{Ptex = Pobj + offset, gText, pointer, dirVec, arrow},
   dirVec = Pobj - Ptex;
   arrow = {Arrowheads[Small], Arrow[{Ptex + 0.2 dirVec, Ptex + 0.8 dirVec}]};
   gText = {Black, Style[Text[lab, Ptex], 10, Italic], arrow}];
DimMarker3[{tail_, head_}, lab_, frac_, offset_] :=
 Module[{labPosn},
  labPosn = tail + frac (head - tail);
  {Arrowheads[{-0.01, 0.01}], Arrow[{tail, head}], Text[lab, labPosn + offset]} ]
```